

Eine Formalisierung des zweiten Satzes von Sylow aus der Gruppentheorie in Naproche im Vergleich zu einer Implementierung in Lean

Moritz Hartlieb, Jonas Lippert

8. März 2020

Inhaltsverzeichnis

1	Einleitung	1
1.1	Theoretische Grundlagen	1
1.2	Vorüberlegungen zur Formalisierung	4
2	Formalisierung in Lean	5
2.1	Quotienten in Lean	5
2.2	Endliche Mengen in Lean	5
2.3	MOD in Lean	7
2.4	Gruppen in Lean	9
2.5	Sylow 2 in Lean	12
3	Formalisierung in Naproche	15
3.1	Struktur	15
3.2	Grundlegende Definitionen	15
3.3	Einführung des Gruppenbegriffs	16
3.4	Untergruppen	17
3.5	Nebenklassen	19
3.6	Zahlen	21
3.7	Primzahlen	24
3.8	Ein Lemma über Potenzen	24
3.9	Endliche Mengen	26
3.10	Der Satz von Lagrange	28
3.11	Gruppenaktionen	30
3.12	Stabilisator und Orbit	31
3.13	Fipunkte	33
3.14	Eine Gruppenaktion	37
3.15	Vorbereitung für den zweiten Satz von Sylow	38
3.16	Der zweite Satz von Sylow	39
4	Vergleich	41
4.1	Direkte Gegenüberstellung	41
4.1.1	Herkömmliche Mathematik	41
4.1.2	Lean	41
4.1.3	Naproche	42
4.2	Beweisführung	43
4.3	Algebraische Strukturen	43
4.4	Funktionsdefinitionen	43
4.5	Struktur	44
4.6	Lesbarkeit	44
4.7	Overloading	44
4.8	User Experience	45

5	Diskussion	46
5.1	Formalisierungskonventionen	46
5.2	Vorschläge zur Weiterentwicklung an Naproche	46
5.3	Lesbarkeit als Ziel	47

1 Einleitung

In der vorliegenden Arbeit stellen wir die Formalisierung des zweiten Sylow-Satzes in der Sprache Lean vor und vergleichen sie mit einer eigenen Formalisierung in Naproche.

Zu Beginn seien die wichtigsten Definitionen und Resultate der Gruppentheorie skizziert auf Basis des Skripts „Eine Einführung in die Algebra (Skript, WS 19/20, Bonn)“ von Prof. Dr. Jan Schröer [1].

1.1 Theoretische Grundlagen

Eine **Gruppe** ist eine Menge G zusammen mit einer Abbildung

$$\circ : G \times G \rightarrow G$$

$$(g, h) \mapsto g \circ h = gh$$

sodass gilt:

- (i) \circ ist assoziativ.
- (ii) Es gibt ein neutrales Element $e \in G$, sodass $e \circ g = g = g \circ e$ für alle $g \in G$.
- (iii) Für alle $g \in G$ existiert ein inverses Element $g^{-1} \in G$, sodass $g \circ g^{-1} = g^{-1} \circ g = e$.

Eine **Untergruppe** H von G , $H \leq G$, ist eine Teilmenge von G mit

- (i) $e \in H$.
- (ii) H ist abgeschlossen bzgl. \circ und Inversenbildung.

Seien $H \leq G$ gegeben. Für $g \in G$ ist

$$gH := \{gh \mid h \in H\}$$

die **Nebenklasse** von H zu g . Es bezeichne G/H die Klasse der Nebenklassen von H . Weiter sei

$$[G : H] := |G/H|$$

der **Index** von H zu G .

Zwei Untergruppen H_1 und H_2 von G sind **konjugiert**, falls es ein $g \in G$ gibt mit

$$H_1 = gH_2g^{-1} := \{g h g^{-1} \mid h \in H\}.$$

Lemma 1.1. Seien $H \leq G$ gegeben. Für $g_1, g_2 \in G$ gilt:

$$(i) \quad g_1H \cap g_2H \neq \emptyset \iff g_1H = g_2H \iff g_1^{-1}g_2 \in H$$

(ii) Für alle $g \in G$ ist die Abbildung

$$H \rightarrow gH$$

$$h \mapsto gh$$

bijektiv. Insbesondere gilt im endlichen Fall $|H| = |gH|$.

□

Lemma 1.2 (Lagrange). Seien $H \leq G$ endlich. Dann folgt aus vorherigem Lemma:

$$|G| = [G : H] \cdot |H|.$$

□

Für eine Gruppe G und eine nichtleere Menge X ist die Abbildung

$$\phi : G \times X \rightarrow X$$

$$(g, x) \mapsto g.x$$

eine **Gruppenaktion**, falls gilt:

- (i) $1.x = x$ für alle $x \in X$.
- (ii) $(gh).x = g.(h.x)$ für alle $g, h \in G$ und $x \in X$.

Weiter definieren wir für $x \in X$:

- (i) den **Orbit** $G.x := \{g.x \mid g \in G\}$ von x ,
- (ii) den **Stabilisator** $G_x := \{g \in G \mid g.x = x\}$ von x ,
- (iii) die Menge der **Fixpunkte** $X^G := \{x \in X \mid g.x = x \text{ für alle } g \in G\}$ von X .

Lemma 1.3. Sei G eine Gruppe und X nichtleer. Die Funktion

$$G/G_x \rightarrow G.x$$

$$(g G_x) \mapsto g.x$$

ist bijektiv.

□

Wie wir in Lemma 1.1 gesehen haben, ist eine Gruppe G disjunkte Vereinigung der Nebenklassen bzgl. einer Untergruppe $H \leq G$. Zusammen mit Lemma 1.3 erhalten wir, dass X disjunkte Vereinigung der Orbits bzgl. einer Gruppenaktion ist und die Kardinalität der Orbits entsprechend Lagrange die Gruppenordnung teilen müssen.

Lemma 1.4 (Bahnenformel). Sei G eine endliche Gruppe, sei $X \neq \emptyset$ endlich und seien x_1, \dots, x_n gegeben, sodass X disjunkte Vereinigung der $G \cdot x_i$ ist. Dann gilt

$$|X| = \sum_{i=1}^n [G : G_{x_i}] = |X^G| + \sum_{\substack{1 \leq i \leq n \\ x_i \notin X^G}} [G : G_{x_i}].$$

□

Falls G eine p -Gruppe ist, das heißt $|G| = p^r$ für eine Primzahl p und ein $r \in \mathbb{N}$, dann folgt:

Lemma 1.5.

$$|X| \equiv |X^G| \pmod{p}.$$

□

Für eine Gruppe G mit $|G| = p^r m$, wobei p prim und $p \nmid m$, ist die Menge der **p -Sylowgruppen** definiert durch

$$\text{Syl}_p(G) := \{P \leq G \mid \}.$$

Wir können nun den zweiten Satz von Sylow formulieren:

Theorem 1.6. Sei p eine Primzahl und G eine endliche Gruppe mit $|G| = p^r \cdot m$, sodass $p \nmid m$. Sei $U \leq G$ eine p -Untergruppe, und sei $P \leq G$ eine p -Sylowgruppe. Dann gilt:

(i) Es gibt ein $g \in G$ mit

$$gUg^{-1} \subseteq P.$$

(ii) Je zwei p -Sylowgruppen sind konjugiert.

Beweis. (i) Setze $X := G/P$ und betrachte die Gruppenaktion

$$\phi : U \times X \rightarrow X$$

$$(u, gP) \mapsto ugP.$$

Nach Lemma 1.5 gilt

$$|X| = [G : P] = m \equiv |X^G| \pmod{p}.$$

Da $p \nmid m$, gilt $X^G \neq \emptyset$. Es existiert also ein $g \in G$, sodass $ugP = gP$ für alle $u \in U$. Folglich ist $g^{-1}Ug \subseteq P$ und damit U konjugiert zu P bzgl. g^{-1} .

(ii) Dies gilt insbesondere im Falle $U \in \text{Syl}_p(G)$.

□

1.2 Vorüberlegungen zur Formalisierung

Es werden also zunächst Grundbegriffe der Gruppentheorie, endliche Mengen sowie natürliche Zahlen, Primzahlen und Modulo-Rechnung benötigt. Hierzu bieten sich unterschiedliche Herangehensweisen an. Es stellt sich heraus, dass Naproche für kleine Theorien gut geeignet ist, deren Grundlagen axiomatisch eingeführt werden, die ihrerseits in einer eigenen Theorie entwickelt werden (können). Die Implementierung in Lean baut hingegen auf bereits formalisierte Grundlagen auf und ist somit Teil einer einzigen großen Theorie der Mathematik.

Interessant ist die Formalisierung von Nebenklassen. Im Sinne einer kleinen Theorie bietet sich in Naproche eine direkte Konstruktion an:

$$\text{Coset}(g, H, G) := \{ g *^G h \mid h \in H \}.$$

Anschließend ist zu zeigen, dass G disjunkte Vereinigung von Nebenklassen bzgl. einer beliebigen Untergruppe H ist. In Lean wird dagegen bzgl. einer Untergruppe S von G folgende Äquivalenzrelation auf G eingeführt:

$$x \sim_S y :\Leftrightarrow x^{-1} * y \in S.$$

Lean erlaubt uns, den Quotient G/\sim_S zu betrachten. Hier wird Lemma 1.1 implizit verwendet.

Im Folgenden wird zunächst auf die jeweiligen Formalisierungen im Detail eingegangen. Anschließend sollen Vor- und Nachteile der jeweiligen Sprachen verglichen und diskutiert werden.

2 Formalisierung in Lean

Die Formalisierung des zweiten Satzes von Sylow in Lean stammt von Chris Hughes[2]. Um mit der aktuellen Version (Stand März 2020) der Leanbibliothek mathlib [3] kompatibel zu sein, musste der Quelltext an einigen Stellen angepasst werden. Die modifizierte Version lässt sich unter <https://github.com/moritz-hl/sylow2> finden.

2.1 Quotienten in Lean

Endliche Mengen und Nebenklassen sind in Lean als Quotient formalisiert. In der core-Library von Lean sind folgende Konstanten definiert:

```
constant quot :  $\Pi$  { $\alpha$  : Sort u}, ( $\alpha \rightarrow \alpha \rightarrow \text{Prop}$ )  $\rightarrow$  Sort u
constant quot.mk :
 $\Pi$  { $\alpha$  : Sort u} (r :  $\alpha \rightarrow \alpha \rightarrow \text{Prop}$ ),  $\alpha \rightarrow$  quot r
axiom quot.ind :
 $\forall$  { $\alpha$  : Sort u} {r :  $\alpha \rightarrow \alpha \rightarrow \text{Prop}$ } { $\beta$  : quot r  $\rightarrow$  Prop},
( $\forall$  a,  $\beta$  (quot.mk r a))  $\rightarrow$   $\forall$  (q : quot r),  $\beta$  q
axiom quot.sound :
 $\forall$  { $\alpha$  : Type u} {r :  $\alpha \rightarrow \alpha \rightarrow \text{Prop}$ } {a b :  $\alpha$ },
r a b  $\rightarrow$  quot.mk r a = quot.mk r b
constant quot.lift :
 $\Pi$  { $\alpha$  : Sort u} {r :  $\alpha \rightarrow \alpha \rightarrow \text{Prop}$ } { $\beta$  : Sort u} (f :  $\alpha \rightarrow \beta$ ),
( $\forall$  a b, r a b  $\rightarrow$  f a = f b)  $\rightarrow$  quot r  $\rightarrow$   $\beta$ 
```

Die Klasse von a in quot r wird durch quot.mk r a erzeugt. Das Induktionsaxiom stellt sicher, dass alle Elemente von quot r von der Form quot.mk r a sind. Die Lifting-Eigenschaft erlaubt es, geeignete Funktionen auf quot r zu liften.

Ein setoid α ist ein Typ α zusammen mit einer Äquivalenzrelation:

```
class setoid ( $\alpha$  : Sort u) :=
(r :  $\alpha \rightarrow \alpha \rightarrow \text{Prop}$ ) (iseqv : equivalence r)
```

Der quotient s auf einem s : setoid α ist dann der Quotient bzgl. einer Äquivalenzrelation:

```
def quotient { $\alpha$  : Sort u} (s : setoid  $\alpha$ ) :=
@quot  $\alpha$  setoid.r
```

Die obigen Eigenschaften von quot werden anschließend auf quotient übertragen.

2.2 Endliche Mengen in Lean

Endliche Mengen werden auf Basis von Listen definiert. Zunächst erhält man Multimengen als Quotienten, indem Permutationen mit Hilfe der Äquivalenzrelation perm miteinander identi-

fiziert werden:

```
inductive perm : list  $\alpha$  → list  $\alpha$  → Prop
| nil    : perm [] []
| skip   :  $\prod$  (x :  $\alpha$ ) {l1 l2 : list  $\alpha$ },
            perm l1 l2 → perm (x::l1) (x::l2)
| swap   :  $\prod$  (x y :  $\alpha$ ) (l : list  $\alpha$ ), perm (y::x::l) (x::y::l)
| trans  :  $\prod$  {l1 l2 l3 : list  $\alpha$ }, perm l1 l2 → perm l2 l3 → perm l1 l3
```

Unter Verwendung des Beweises `perm.eqv`, dass `perm` eine Äquivalenzrelation ist, wird eine Instanz von `setoid list α` eingeführt, welche dann in der Definition der Multimenge Verwendung findet:

```
instance is_setoid ( $\alpha$ ) : setoid (list  $\alpha$ ) :=
  setoid.mk (@perm  $\alpha$ ) (perm.eqv  $\alpha$ )
```

```
def {u} multiset ( $\alpha$  : Type u) : Type u :=
  quotient (list.is_setoid  $\alpha$ )
```

Eine Multimenge wird zur endlichen Menge, wenn sie keine Duplikate beinhaltet:

```
structure finset ( $\alpha$  : Type*) :=
  (val : multiset  $\alpha$ )
  (nodup : nodup val)
```

Die Definition von `nodup` für Multimengen basiert auf `nodup` für Listen.

```
def nodup : list  $\alpha$  → Prop := pairwise (≠)
```

Hierbei prüft `pairwise`, ob die Relation \neq paarweise gilt.

```
variables (R :  $\alpha$  →  $\alpha$  → Prop)
inductive pairwise : list  $\alpha$  → Prop
| nil {} : pairwise []
| cons :  $\forall$  {a :  $\alpha$ } {l : list  $\alpha$ }, ( $\forall$  a' ∈ l, R a a') → pairwise l →
  pairwise (a::l)
```

Jetzt wird `nodup` für Listen auf Multimengen geliftet. Dazu werden `quot.lift_on` folgende Parameter übergeben: der Quotient `s`, die zu liftende Funktion `nodup` und ein Beweis, dass Permutation keine Duplikate erzeugt: Aus den zwei Listen `s` `t` und dem Beweis `p`, dass diese in Relation bzgl. `perm` zueinander stehen, erzeugt `perm_nodup` die Äquivalenz `nodup s \Leftrightarrow nodup t`. Dann wird das Lean-interne Axiom `propext` verwendet, nach dem äquivalente Propositionen gleich sind. (Die Rechtsklammerung bei der Funktionseinsetzung wird hier durch `$` gewährleistet.)

```

def nodup (s : multiset  $\alpha$ ) : Prop :=
quot.lift_on s nodup ( $\lambda$  s t p, propext $ perm_nodup p)

def quot.lift_on { $\alpha$  : Sort u} { $\beta$  : Sort v} {r :  $\alpha \rightarrow \alpha \rightarrow$  Prop}
(q : quot r) (f :  $\alpha \rightarrow \beta$ ) (c :  $\forall$  a b, r a b  $\rightarrow$  f a = f b) :  $\beta$ 

theorem perm_nodup {l1 l2 : list  $\alpha$ } : l1 ~ l2  $\rightarrow$ 
(nodup l1  $\leftrightarrow$  nodup l2)

```

Der Beweis von perm_nodup kann in perm.lean nachgelesen werden. In der Regel wird nicht finset direkt verwendet, sondern die Typenklasse fintype, sodass der Typ α selber endlich ist.

```

class fintype ( $\alpha$  : Type*) :=
(elems : finset  $\alpha$ )
(complete :  $\forall$  x :  $\alpha$ , x  $\in$  elems)

```

2.3 MOD in Lean

Die Definition von $x \pmod y$ auf den natürlichen Zahlen basiert auf der Wohlfundiertheit der $<$ -Relation auf \mathbb{N} :

```

inductive acc { $\alpha$  : Sort u} (r :  $\alpha \rightarrow \alpha \rightarrow$  Prop) :  $\alpha \rightarrow$  Prop
| intro (x :  $\alpha$ ) (h :  $\forall$  y, r y x  $\rightarrow$  acc y) : acc x

parameters { $\alpha$  : Sort u} {r :  $\alpha \rightarrow \alpha \rightarrow$  Prop}

local infix '<':50      := r

inductive well_founded { $\alpha$  : Sort u} (r :  $\alpha \rightarrow \alpha \rightarrow$  Prop) : Prop
| intro (h :  $\forall$  a, acc r a) : well_founded

class has_well_founded ( $\alpha$  : Sort u) : Type u :=
(r :  $\alpha \rightarrow \alpha \rightarrow$  Prop) (wf : well_founded r)

```

Ein Element $x : \alpha$ kann nur die Eigenschaft acc haben, falls ein „kleinstes“ Element bzgl. der Relation R existiert. Entsprechend wird ein Rekursions- und Induktionsprinzip für wohlfundierte Relationen eingeführt, auf die hier nicht näher eingegangen werden soll. Wichtig ist der Spezialfall, dass sich über wohlfundierte Relationen Funktionen definieren lassen (siehe def fix weiter unten):

```

parameter hwf : well_founded r
variable {C :  $\alpha \rightarrow$  Sort v}

```

```
variable F :  $\Pi x, (\Pi y, y < x \rightarrow C y) \rightarrow C x$ 
```

```
def fix_F (x :  $\alpha$ ) (a : acc r x) : C x :=  
acc.rec_on a ( $\lambda x_1 ac_1 ih, F x_1 ih$ )
```

Die Konstruktion des Fixpunktes nimmt gemäß `acc.rec_on` einen Beweis `a` für `acc r x` und liefert `C x`, falls folgender Sachverhalt gegeben ist:

$$(\Pi (x_1 : \alpha), (\forall (y : \alpha), r y x_1 \rightarrow \text{acc } r y) \rightarrow \\ (\Pi (y : \alpha), r y x_1 \rightarrow C y) \rightarrow C x_1)$$

Durch `acc.rec_on` erhalten wir über den Konstruktor `intro`:

```
x1 :  $\alpha$   
ac1 :  $\forall (y : \alpha), r y x_1 \rightarrow \text{acc } r y$   
ih :  $\Pi (y : \alpha), r y x_1 \rightarrow C y$ 
```

`F` liefert das Gewünschte. Wir können die Fixpunkteigenschaft nun auf wohlfundierte Relationen übertragen:

```
variables { $\alpha$  : Sort u} {C :  $\alpha \rightarrow$  Sort v} {r :  $\alpha \rightarrow \alpha \rightarrow \text{Prop}$ }  
  
def fix (hwf : well_founded r) (F :  $\Pi x, (\Pi y, r y x \rightarrow C y) \rightarrow C x$ ) (  
  x :  $\alpha$ ) : C x :=  
fix_F F x (apply hwf x)
```

Ist eine Funktion `F` gegeben, welche die Werte `C x` für bekannte Werte `C y` der „kleineren“ Elemente `y < x` liefert, so ist nach `fix` die Funktion `C` für alle `x : α` definiert. Für `F` setzen wir

```
private def mod.F (x : nat) (f :  $\Pi x_1, x_1 < x \rightarrow \text{nat} \rightarrow \text{nat}$ ) (y : nat)  
  : nat :=  
if h :  $0 < y \wedge y \leq x$  then f (x - y) (div_rec_lemma h) y else x
```

und damit `C` auf $\mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$. Hier wurde folgendes Lemma verwendet:

```
div_rec_lemma {x y : nat} :  $0 < y \wedge y \leq x \rightarrow x - y < x$ .
```

Weiter können wir `mod` und damit eine Instanz der Typenklasse `class has_mod` definieren, die bereits in `core.lean` vordefiniert ist. Sie besitzt einzig den Konstruktor `has_mod.mod : $\mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$` .

```
protected def mod := fix lt_wf mod.F
```

```
instance : has_mod nat :=
⟨nat.mod⟩
```

Die Infixnotation für `has_mod.mod n m` ist `n % m`. Schließlich wird `a ≡ b [MOD n]` definiert:

```
def modeq (n a b : ℕ) := a % n = b % n

notation a ' ≡ '[:50] b ' [MOD '[:50] n ']'[:0] := modeq n a b
```

2.4 Gruppen in Lean

Gruppen werden sukzessive durch Erweiterungen von `type classes` definiert:

```
class has_mul (α : Type u) := (mul : α → α → α)

infix * := has_mul.mul

class semigroup (α : Type u) extends has_mul α :=
(mul_assoc : ∀ a b c : α, a * b * c = a * (b * c))

class monoid (α : Type u) extends semigroup α, has_one α :=
(one_mul : ∀ a : α, 1 * a = a) (mul_one : ∀ a : α, a * 1 = a)

class group (α : Type u) extends monoid α, has_inv α :=
(mul_left_inv : ∀ a : α, a-1 * a = 1)
```

Entsprechendes gilt für Untergruppen

```
variables {α : Type*} [monoid α] {s : set α}

class is_submonoid (s : set α) : Prop :=
(one_mem : (1:α) ∈ s)
(mul_mem {a b} : a ∈ s → b ∈ s → a * b ∈ s)

class is_subgroup (s : set α) extends is_submonoid s : Prop :=
(inv_mem {a} : a ∈ s → a-1 ∈ s)
```

und Gruppenaktionen:

```
class has_scalar (α : Type u) (γ : Type v) :=
(smul : α → γ → γ)

infixr ' · '[:73] := has_scalar.smul
```

```

class mul_action ( $\alpha$  : Type u) ( $\beta$  : Type v) [monoid  $\alpha$ ] extends
  has_scalar  $\alpha$   $\beta$  :=
(one_smul :  $\forall b : \beta, (1 : \alpha) \cdot b = b$ )
(mul_smul :  $\forall (x\ y : \alpha) (b : \beta), (x * y) \cdot b = x \cdot y \cdot b$ )

```

Es folgen die üblichen Definitionen bzgl. Gruppenaktionen.

```

variables ( $\alpha$ ) [monoid  $\alpha$ ] [mul_action  $\alpha$   $\beta$ ]

def orbit (b :  $\beta$ ) := set.range ( $\lambda x : \alpha, x \cdot b$ )

variables ( $\alpha$ ) ( $\beta$ )

def stabilizer (b :  $\beta$ ) : set  $\alpha$  :=
{x :  $\alpha$  |  $x \cdot b = b$ }

def fixed_points : set  $\beta$  := {b :  $\beta$  |  $\forall x, x \in \text{stabilizer } \alpha\ b$ }

```

Nebenklassen werden als Quotient bzgl. der Relation left_rel definiert:

```

def left_rel [group  $\alpha$ ] (s : set  $\alpha$ ) [is_subgroup s] : setoid  $\alpha$  :=
< $\lambda x\ y, x^{-1} * y \in s$ ,
  assume x, by simp [is_submonoid.one_mem],
  assume x y hxy,
  have  $(x^{-1} * y)^{-1} \in s$ , from is_subgroup.inv_mem hxy,
  by simpa using this,
  assume x y z hxy hyz,
  have  $x^{-1} * y * (y^{-1} * z) \in s$ , from is_submonoid.mul_mem hxy hyz,
  by simpa [mul_assoc] using this>

def left_cosets [group  $\alpha$ ] (s : set  $\alpha$ ) [is_subgroup s] : Type* :=
  quotient (left_rel s)

```

Für eine Untergruppe H von G ist dann left_rel H ein setoid G mit entsprechender Relation und dem zugehörigen Beweis, dass es sich um eine Äquivalenzrelation handelt. Mithilfe dieser Definition wird eine Instanz von fintype bzgl. left_costes H erstellt:

```

noncomputable instance [fintype G] (H : set G) [is_subgroup H] :
  fintype (left_cosets H) :=
  quotient.fintype (left_rel H)

```

Der Grund dafür, dass die Instanz als noncomputable markiert werden muss, ist die Verwendung von decidable.eq in quotient.fintype. Das geht auf den Umstand zurück,

dass das Bild einer Funktion auf einem endlichen Typ wieder ein endlicher Typ β ist. Hierbei wird das Bild zunächst als Multimenge betrachtet. Anschließend werden eventuelle Duplikate durch den Operator `to_finset` entfernt, was die Entscheidbarkeit der Gleichheitsrelation auf β voraussetzt. Um den 2. Sylowsatz in Lean zu formulieren, fehlen noch die Definitionen des `conjugate_set` und der p -Sylowgruppen:

```
def conjugate_set (x : G) (H : set G) : set G :=
  (λ n, x⁻¹ * n * x) ⁻¹' H

class is_sylow [fintype G] (H : set G) {p : ℕ} (hp : prime p) extends
  is_subgroup H : Prop :=
  (card_eq : card H = p ^ dlogn p (card G))

lemma sylow_2 [fintype G] {p : ℕ} (hp : nat.prime p)
  (H K : set G) [is_sylow H hp] [is_sylow K hp] :
  ∃ g : G, H = conjugate_set g K
```

`dlogn p (card G)` ist die Vielfachheit von p in `card G`.

`#print axioms sylow.sylow_2` zeigt, dass `propext`, `quot.sound` und `classical.choice` verwendet wird.

`quot.sound` wird benötigt, um zu beweisen, dass das kanonische Operieren einer Untergruppe H auf den Nebenklassen einer Untergruppe K eine Gruppenaktion ist:

```
def mul_left_cosets (L₁ L₂ : set G) [is_subgroup L₂] [is_subgroup L₁]
  (x : L₂) (y : left_cosets L₁) : left_cosets L₁ :=
  quotient.lift_on y (λ y, [(x : G) * y])
  (λ a b (hab : _ ∈ L₁), quotient.sound
    (show _ ∈ L₁, by rwa [mul_inv_rev, ← mul_assoc, mul_assoc (a⁻¹),
      inv_mul_self, mul_one])))
```

Dafür soll die Funktion $(\lambda y, [(x : G) * y])$ auf Nebenklassen geliftet werden. Für zwei Elemente a und b , die bzgl. `left_rel` äquivalent sind, ist also zu zeigen, dass

$$[(x : G) * a] = [(x : G) * b].$$

Mit `quot.sound` genügt dann ein Beweis dafür, dass die erzeugenden Elemente äquivalent sind. Die nötigen Umformungen erledigt `simp`.

Im Beweis von Sylow 2 werden zunächst die vorangegangenen Resultate verwendet, um herzuleiten, dass die Anzahl der Fixpunkte bzgl. obiger Aktion ungleich Null ist. An dieser Stelle liefert `classical.choice` einen solchen Fixpunkt, ohne den die Konstruktion eines geeigneten `conjugate_set` nicht möglich wäre.

2.5 Sylow 2 in Lean

Wir wollen nun auf einige Details im Beweis des Satzes eingehen. Die verwendeten Lemmata, die für das sukzessive Umformen via `rw` oder `simp` genutzt werden, sind dabei weder mathematisch noch hinsichtlich des Formalisierungsprozesses besonders interessant. Es werden zahlreiche technische Lemma über natürliche Zahlen, Kardinalitäten, Funktionen, etc. zusammengetragen, deren Beweise für den Menschen trivial und in Lean im Nachhinein teilweise nur sehr schwer verständlich sind. Es ist prinzipiell klar, dass es auf Basis der konstruierten Begriffe und Strukturen möglich ist, entsprechende Beweise für Lean verständlich zu formulieren und es genügt in der Regel zu wissen, dass sie existieren.

Die ersten drei Hilfslemma liefern uns den gewünschten Fixpunkt.

```
lemma sylow_2 [fintype G] {p : ℕ} (hp : nat.prime p)
(H K : set G) [is_sylow H hp] [is_sylow K hp] :
  ∃ g : G, H = conjugate_set g K :=

have hs : card (left_cosets K) = card G / (p ^ dlogn p (card G)) :=
  (nat.mul_right_inj (pos_pow_of_pos (dlogn p (card G)) hp.pos)).1
$ by rw [← card_sylow K hp, ← card_eq_card_cosets_mul_card_subgroup,
  card_sylow K hp,
  nat.div_mul_cancel (dlogn_dvd _ hp.1)],

have hmodeq : card G / (p ^ dlogn p (card G)) ≡ card (fixed_points H (
  left_cosets K)) [MOD p] :=
eq.subst hs (mul_action.card_modeq_card_fixed_points hp (card_sylow H
  hp)),

have hfixed : 0 < card (fixed_points H (left_cosets K)) :=
nat.pos_of_ne_zero
(λ h, (not_dvd_div_dlogn (fintype.card_pos_iff.2 ⟨(1 : G)⟩) hp.1)
  (by rwa [h, nat.modeq.modeq_zero_iff] at hmodeq)),
```

Zunächst wird in `hs` die Kardinalität der Klasse der Nebenklassen auf Basis der Definition von Sylow-Gruppen umgeformt. Hierbei wird das Lemma von Lagrange unter dem Namen `card_eq_card_cosets_mul_card_subgroup` benutzt.

In `hmodeq` kommt die Lean-Version von Lemma (??) zur Anwendung:

```
lemma card_modeq_card_fixed_points [fintype α] [fintype G]
[fintype (fixed_points G α)]
{p n : ℕ} (hp : nat.prime p) (h : card G = p ^ n) :
card α ≡ card (fixed_points G α) [MOD p]
```

Durch Substitution wird das allgemeine Resultat auf den Spezialfall übertragen.

Mit Hilfe von `hmodeq` lässt sich schließlich `hfixed` zeigen, dass nämlich die Kardinalität der Fixpunkte größer Null ist. Nach `nat.pos_of_ne_zero` genügt es zu zeigen, dass die

Kardinalität der Fixpunkte ungleich Null ist. Es wird also ein Beweis von `false` gefordert unter Annahme `h`, die Kardinalität sei gleich Null. Der Widerspruch wird erzeugt, in dem erst

```
lemma not_dvd_div_dlogn {p a : ℕ} (ha : a > 0) (hp : p > 1) :
  ¬p | a / (p ^ dlogn p a)
```

auf `a = card G` angewendet wird. Dass `p` aber ein Teiler der rechten Seite sein muss, folgt aus der Annahme `h` zusammen mit

```
modeq_zero_iff : b ≡ 0 [MOD n] ↔ n | b,
```

wobei `b = card G / (p ^ dlogn p a)`.

Um aus dem abstrakten Kardinalitäts-Argument einen konkreten Fixpunkt zu gewinnen, wird das Axiom `classical.choice` in Form des Lemmas `fintype.card_pos_iff` verwendet:

```
let ⟨x, hx⟩ := fintype.card_pos_iff.1 hfixed in

begin

revert hx
```

Die Taktik `revert` generalisiert den Beweis `hx`, dass `x` ein Fixpunkt ist, sodass nun folgendes zu beweisen ist:

```
x ∈ fixed_points H (left_cosets K) →
(∃ (g : G), H = conjugate_set g K)
```

Der Grund dafür ist der, dass wir für die weitere Argumentation einen Vertreter `g : G` für die Äquivalenzklasse `x` benötigen. Dies wird möglich durch das Axiom `quot.ind` in Form von `quotient.induction_on`, nachdem es genügt, das Ziel für einen beliebigen Vertreter zu zeigen:

```
refine quotient.induction_on x
(λ g hg, ⟨g, set.eq_of_card_eq_of_subset _ _⟩)
```

Für ein Element `g : G` und unter der Annahme `hg`, dass `[[g]]` ein Fixpunkt ist, bleibt nun die ursprünglich Aussage `∃(g : G), H = conjugate_set g K` zu zeigen. Dem anonymen Konstruktor wurde zur Konstruktion der Existenzaussage das Element `g` und folgendes Lemma übergeben:

```
lemma eq_of_card_eq_of_subset {s t : set α} [fintype s] [fintype t]
(hcard : card s = card t) (hsub : s ⊆ t) : s = t
```


Durch die Verwendung der Taktik `refine` und der zwei Unterstriche anstelle der Beweise `hcard` und `hsub`, werden diese Bedingungen als neue Ziele innerhalb des Taktik-Blocks erstellt, wobei das ursprüngliche Ziel als bewiesen gilt. Die beiden Beweise seien hier der Vollständigkeit halber aufgeführt, auch wenn die technischen Details nicht weiter interessant sind:

```
{
rw [conjugate_set_eq_image, set.card_image_of_injective _
    conj_inj_left,
card_sylow K hp, card_sylow H hp] },
{
assume y hy,
have : (y⁻¹ * g)⁻¹ * g ∈ K :=
quotient.exact ((mem_fixed_points' (left_cosets K)).1 hg [[y⁻¹ * g]]
⟨⟨y⁻¹, inv_mem hy⟩, rfl⟩),
simp [conjugate_set_eq_preimage],
simp only [*, mul_assoc, mul_inv_rev] at *,
simp [*, inv_inv] at *}
end
```

3 Formalisierung in Naproche

3.1 Struktur

Die Formalisierung in Naproche teilt sich in 8 Dateien auf:

Name	Inhalt	Abschnitt
01basicgrouptheory.ftl	Gruppentheorie	3.2
02numbers.ftl	Zahlen	3.6
03cards.ftl	Kardinalitäten	3.9
04lagrange.ftl	Satz von Lagrange	3.10
05statorb.ftl	Gruppenaktionen	3.11
06fixedpointsmodp.ftl	Fixpunkte	3.13
07grpaction.ftl	SylowGruppenaktion	3.14
08syllow2.ftl	Der Satz von Sylow	3.15

In jeder Datei werden zu Beginn jeweils die Aussagen und Definitionen der vorangegangenen Dateien eingebunden. Hierfür wurden Hilfsdateien `xxExport.ftl` erstellt, in denen zu jeder Datei die daran enthaltenen Definitionen und Theoreme in Form von Axiomen ohne Beweise festgehalten wurden. Grundsätzlich wäre es möglich diese Dateien über den Befehl `[read dateiname.ftl]` einzulesen, da dieser Befehl jedoch in der jetzigen Version von Naproche nicht auf allen Betriebssystemen gleichermaßen funktioniert, haben wir die Dateien manuell kopiert und dies durch Kommentare im Quelltext vermerkt.

Die Formalisierung ließ sich auf einem MacBook Pro (16 GB RAM, Intel i7 2,7 GhZ) von 2018 innerhalb der Isabelle-Naproche Umgebung in rund 5 Minuten verifizieren. Fügt man jedoch alle Dateien aneinander, ohne die Theoreme jeweils durch Axiome zu ersetzen, dann ist Naproche nicht mehr in der Lage, den Beweis vollständig zu verifizieren.

Der Dateien finden sich unter <https://github.com/moritz-hl/syllowftl>.

3.2 Grundlegende Definitionen

Zunächst führen wir einige grundlegenden Definitionen ein. Interessant ist hier insbesondere der Begriff der `disjunct collection` den wir einführen um später über disjunkte Vereinigung "sprechen" zu können.

`Let M, N denote sets.`

`Let x << M` stand for `x` is an element of `M`.

Definition.

`Prod(M,N) = { (x,y) | x << M and y << N }.`

Definition.

A subset of `M` is a set `N` `such that` every element of `N` is an element of `M`.

Definition.

`Let M` be a set.

`M` is empty iff there is no element `x` of `M` `such that` `x = x`.

Definition.

Let M be a set **such that** for all elements N of M N is a set.

$\backslash-/ M = \{x \mid \text{There is an element } N \text{ of } M \text{ such that } x \text{ is an element of } N\}.$

Definition.

Let N_1, N_2 be sets.

$N_1 \backslash-/ N_2 = \{x \mid x \text{ is an element of } N_1 \text{ or } x \text{ is an element of } N_2\}.$

Definition.

Let N_1 be a set.

Let N_2 be a subset of N_1 .

$N_1 \backslash \backslash N_2 = \{x \mid x \text{ is an element of } N_1 \text{ and } (x \text{ is not an element of } N_2)\}.$

Definition.

Let N_1, N_2 be a sets.

N_1 and N_2 are disjunct iff there is no element x of N_1 **such that** x is an element of N_2 .

Definition.

A disjunct collection is a set M **such that**

(for all elements N of M N is a set) and for all elements N_1, N_2 of M ($N_1 = N_2$ or (N_1 and N_2 are disjunct)).

Definition.

Let f be a function. Let M, N be sets. f is from M to N iff $\text{Dom}(f) = M$ and for every element x of M $f[x]$ is an element of N .

Definition.

Let f be a function. $\text{Range}(f) = \{f[x] \mid x \ll \text{Dom}(f)\}.$

Definition.

Let f be a function. f is injective iff for all elements x, y of $\text{Dom}(f)$ **we have** ($x \neq y \Rightarrow f[x] \neq f[y]$).

Definition.

Let f be a function. f is surjective onto M iff

(f is from $\text{Dom}(f)$ to M

and for every $y \ll M$ there is an element x of $\text{Dom}(f)$ **such that** $f[x]=y$).

3.3 Einführung des Gruppenbegriffs

Der Gruppenbegriff wird abstrakt als Notion eingeführt. Für diese Notion werden die Existenz einer Trägermenge $E1(G)$, die Existenz der Eins, Inversen und Verknüpfung sowie deren Eigenschaften axiomatisch gefordert. Die Verknüpfung $a *^G b$ muss hier explizit durch G gekennzeichnet werden.

[synonym group/-s]
[synonym subgroup/-s]

Signature.

A group is a notion.

Let G denote a group.

Signature.

$\text{El}(G)$ is a set.

Signature.

$\text{One}(G)$ is an object.

Axiom.

$\text{One}(G) \ll \text{El}(G)$.

Signature.

Let a, b be elements of $\text{El}(G)$.

$a *^{\{G\}} b$ is an element of $\text{El}(G)$.

Signature.

Let a be an element of $\text{El}(G)$.

$\text{Inv}(a, G)$ is an element of $\text{El}(G)$.

Axiom Assoc.

Let x, y, z be elements of $\text{El}(G)$. $x *^{\{G\}} (y *^{\{G\}} z) = (x *^{\{G\}} y) *^{\{G\}} z$.

Axiom InvOne.

Let x be an element of $\text{El}(G)$. $x *^{\{G\}} \text{Inv}(x, G) = \text{One}(G) = \text{Inv}(x, G) *^{\{G\}} x$.

Axiom MulOne.

Let x be an element of $\text{El}(G)$. $x *^{\{G\}} \text{One}(G) = x = \text{One}(G) *^{\{G\}} x$.

Lemma InvUniq.

Let x, y be elements of $\text{El}(G)$.

If $x *^{\{G\}} y = \text{One}(G)$ then $y = \text{Inv}(x, G)$.

Lemma OneUniq.

Let x, y be elements of $\text{El}(G)$.

If $x *^{\{G\}} y = x$ then $y = \text{One}(G)$.

3.4 Untergruppen

Untergruppen werden als Teilmengen von $\text{El}(G)$ eingeführt. Im Beweis von Sylow wird genutzt, dass Untergruppen selbst wieder eine Gruppenstruktur annehmen. Da wir den Begriff der Gruppe abstrakt eingeführt haben, ist es hier nicht möglich, zu beweisen, dass $\text{Gr}(U, G)$

eine Gruppe ist, weswegen dies hier auch als Axiom gefordert wird.

Definition.

A subgroup of G is set H such that
 $(H \text{ is a subset of } \text{El}(G))$
and $(\text{One}(G) \ll H)$
and (for every $x \ll H$ $\text{Inv}(x, G) \ll H$)
and (for all elements x, y of H $x \cdot^G y \ll H$).

Definition.

Let U be a subgroup of G .
 $\text{Gr}(U, G)$ is a group H such that
 $(\text{El}(H) = U)$
and $(\text{One}(H) = \text{One}(G))$
and (for every $x \ll U$ $\text{Inv}(x, H) = \text{Inv}(x, G)$)
and (for all elements x, y of U $x \cdot^{\text{Gr}(U, G)} y = x \cdot^G y$).

Lemma.

Let G be a group.

Let H be a subset of $\text{El}(G)$.

Assume $((\text{There is a } x \ll H \text{ such that } x = x) \text{ and } (\text{for all elements } y, z \text{ of } H \text{ } z \cdot^G y \ll H))$.

H is a subgroup of G .

Proof.

$\text{One}(G) \ll H$.

Proof.

Take $x \ll H$.

Then $\text{One}(G) = x \cdot^G \text{Inv}(x, G)$.

Thus $\text{One}(G) \ll H$.

end.

For every $x \ll H$ $\text{Inv}(x, G) \ll H$.

Proof.

Let x be an element of H .

Then $\text{Inv}(x, G) = \text{One}(G) \cdot^G \text{Inv}(x, G)$.

Thus $\text{Inv}(x, G) \ll H$.

end.

For all elements x, y of H $x \cdot^G y \ll H$.

Proof.

Let x, y be elements of H .

Then $\text{Inv}(x, G) \ll H$.

$x \cdot^G y = x \cdot^G \text{Inv}(\text{Inv}(y, G), G)$.

Hence $x \cdot^G y \ll H$.

end.

Qed.

3.5 Nebenklassen

Wir führen Nebenklassen im Gegensatz zum Ansatz der Formalisierung in Lean nicht über Quotienten ein, sondern über die mengenbasierte Definition. Das in der Einführung genannte Lemma begründet, dass diese Ansätze das gleiche metatheoretische Objekt formalisieren.

Definition.

Let g be an element of $\text{El}(G)$.

Let H be a subgroup of G .

$\text{Coset}(g, H, G) = \{g *^{\{G\}} h \mid h \ll H\}$.

Lemma.

Let H be a subgroup of G .

Let g_1, g_2 be elements of $\text{El}(G)$.

Assume $\text{Coset}(g_1, H, G)$ and $\text{Coset}(g_2, H, G)$ are not disjoint.

$\text{Inv}(g_2, G) *^{\{G\}} g_1 \ll H$.

Proof.

Take $y \ll \text{El}(G)$ such that $(y \ll \text{Coset}(g_1, H, G) \text{ and } y \ll \text{Coset}(g_2, H, G))$.

Take $b \ll H$ such that $y = g_1 *^{\{G\}} b$.

Take $c \ll H$ such that $y = g_2 *^{\{G\}} c$.

We have $g_1 = y *^{\{G\}} \text{Inv}(b, G)$.

$g_2 = y *^{\{G\}} \text{Inv}(c, G)$.

$\text{Inv}(g_2, G) = c *^{\{G\}} \text{Inv}(y, G)$.

$\text{Inv}(y, G) *^{\{G\}} g_1 = \text{Inv}(b, G)$.

Therefore $\text{Inv}(g_2, G) *^{\{G\}} g_1 = c *^{\{G\}} (\text{Inv}(b, G))$.

qed.

Lemma.

Let H be a subgroup of G .

Let g_1, g_2 be elements of $\text{El}(G)$.

If $\text{Inv}(g_2, G) *^{\{G\}} g_1 \ll H$

Then $\text{Coset}(g_1, H, G) = \text{Coset}(g_2, H, G)$.

Proof.

Assume $\text{Inv}(g_2, G) *^{\{G\}} g_1 \ll H$.

Every element of $\text{Coset}(g_1, H, G)$ is an element of $\text{Coset}(g_2, H, G)$.

Proof.

Let y be an element of $\text{Coset}(g_1, H, G)$.

Take $a \ll H$ such that $y = g_1 *^{\{G\}} a$.

$(\text{Inv}(g_2, G) *^{\{G\}} g_1) *^{\{G\}} a \ll H$.

Then $y = g_1 *^{\{G\}} a = g_2 *^{\{G\}} ((\text{Inv}(g_2, G) *^{\{G\}} g_1) *^{\{G\}} a)$.

end.

Every element of $\text{Coset}(g_2, H, G)$ is an element of $\text{Coset}(g_1, H, G)$.

Proof.

Let y be an element of $\text{Coset}(g_2, H, G)$.
 Take $a \in H$ such that $y = g_2 *_{\{G\}} a$.
 $(\text{Inv}(g_2, G) *_{\{G\}} g_1) *_{\{G\}} a \in H$.
 Then $y = g_2 *_{\{G\}} a = g_1 *_{\{G\}} ((\text{Inv}(g_1, G) *_{\{G\}} g_2) *_{\{G\}} a)$.
 end.

Therefore $\text{Coset}(g_1, H, G) = \text{Coset}(g_2, H, G)$.
 Qed.

Lemma CosEq.

Let H be a subgroup of G .
 Let g_1, g_2 be elements of $\text{El}(G)$.
 If $\text{Coset}(g_1, H, G)$ and $\text{Coset}(g_2, H, G)$ are not disjoint
 then $\text{Coset}(g_1, H, G) = \text{Coset}(g_2, H, G)$.

Lemma.

Let H be a subgroup of G .
 Let g_1, g_2 be elements of $\text{El}(G)$.
 $\text{Inv}(g_2, G) *_{\{G\}} g_1 \in H$ iff $\text{Coset}(g_1, H, G) = \text{Coset}(g_2, H, G)$.

Definition.

Let H be a subgroup of G .
 $\text{Cosets}(H, G) = \{\text{Coset}(g, H, G) \mid g \in \text{El}(G)\}$.

[synonym coset/-s]

Let a coset of H in G denote an element of $\text{Cosets}(H, G)$.

Lemma.

Let U be a subgroup of G .
 $\text{El}(G) = \bigcup \text{Cosets}(U, G)$.
 Proof.
 Let us show that every element of $\text{El}(G)$ is an element of $\bigcup \text{Cosets}(U, G)$.
 Let g be an element of $\text{El}(G)$.
 g is an element of $\text{Coset}(g, U, G)$.
 end.

Let us show that every element of $\bigcup \text{Cosets}(U, G)$ is an element of $\text{El}(G)$.
 Let h be an element of $\bigcup \text{Cosets}(U, G)$.
 Take an element k of $\text{El}(G)$ such that h is an element of $\text{Coset}(k, U, G)$.
 $\text{Coset}(k, U, G)$ is a subset of $\text{El}(G)$.
 Hence h is an element of $\text{El}(G)$.
 end.

Therefore $\text{El}(G) = \bigcup \text{Cosets}(U, G)$.
 Qed.

Lemma.

Let G be a group.

Let U be a subgroup of G .
 $\text{Cosets}(U, G)$ is a disjunct collection.
Proof.
 Let us show that for every elements $N1, N2$ of $\text{Cosets}(U, G)$ $N1 = N2$ or ($N1$ and $N2$ are disjunct).
 Let $N1, N2$ be cosets of U in G .
 Take elements $g1, g2$ of $\text{El}(G)$ such that $N1 = \text{Coset}(g1, U, G)$ and $N2 = \text{Coset}(g2, U, G)$.
 If $N1$ and $N2$ are not disjunct then $N1 = N2$ (by CosEq).
 Therefore the thesis.
 end.
 Qed.

Lemma.
 Let G be a group.
 Let U be a subgroup of G .
 U is a coset of U in G .
Proof.
 We have $U = \text{Coset}(\text{One}(G), U, G)$.
 Therefore the thesis.
 Qed.

Definition.
 Let g be an element of $\text{El}(G)$.
 Let U be a subgroup of G .
 $\text{Conjugate}(g, U, G) = \{(g * \{G\} (u * \{G\} \text{Inv}(g, G))) \mid u \text{ is an element of } U\}$.

Definition.
 Let U, V be subgroups of G .
 U and V are conjugates in G iff there is an element g of $\text{El}(G)$ such that $U = \text{Conjugate}(g, V, G)$.

3.6 Zahlen

Da wir uns mit endlichen Gruppen, also insbesondere mit Kardinalitäten beschäftigen, führen wir nun ganze Zahlen axiomatisch ein. Übernommen : Interessant, während MulZero , MulMinOne in der Beispieldatei nicht weiter bewiesen werden mussten, musste auf.

Ein herkömmlicher mathematischer Text über die Sylowsätze behandelt auch nicht elementare Zahlentheorie sondern nimmt eine gewisse Vorkenntnis des Lesers an.

In diesem Sinne führen wir auch die Zahlen abstrakt ein und fordern wohlbekannte Eigenschaften axiomatisch.

[synonym integer/-s]
 [synonym number/-s]

Signature Integers. An integer is a notion.

Signature Naturals. A natural number is an integer.

Let a, b, c, d, e, n, m stand for integers.

Signature NatZero. 0 is a natural number.

Signature NatOne. 1 is a natural number.

Signature IntNeg. $-a$ is an integer.

Signature IntPlus. $a + b$ is an integer.

Signature IntMult. $a * b$ is an integer.

Signature NatPot. Let b be a natural number. $a ^ b$ is an integer.

Axiom NatPlus. If a and b are natural numbers then $a + b$ is a natural number.

Axiom NatMult. If a and b are natural numbers then $a * b$ is a natural number.

Signature NatLT. $a < b$ is an atom.

Let a is smaller than b stand for $a < b$.

Axiom TriCh.

$a = b \vee a < b \vee b < a$.

Axiom.

$a < b$ iff $a \neq b$.

Let $a - b$ stand for $a + (-b)$.

Axiom NatSub.

If $a < b$ then $b - a$ is natural number.

Axiom AddAsso. $a + (b + c) = (a + b) + c$.

Axiom AddComm. $a + b = b + a$.

Axiom AddZero. $a + 0 = a = 0 + a$.

Axiom AddNeg. $a - a = 0 = -a + a$.

Axiom MulAsso. $a * (b * c) = (a * b) * c$.

Axiom MulComm. $a * b = b * a$.

Axiom MulOne. $a * 1 = a = 1 * a$.

Axiom Distrib. $a * (b + c) = (a*b) + (a*c)$ and
 $(a + b) * c = (a*c) + (b*c)$.

Axiom ZeroDiv. $a \neq 0 \wedge b \neq 0 \Rightarrow a * b \neq 0$.

Axiom PotInj. Let p be an integer. Let n, m be natural numbers. $(p ^ n = p ^ m) \Rightarrow n = m$.

Axiom PotAdd. Let p be an integer. Let n, m be natural numbers. $p ^ (n + m) = (p ^ n) * (p ^ m)$.

Axiom PotNotZero. Let p be an integer. Let k be a natural number. $p^k \neq 0$.

Lemma MulZero. $a * 0 = 0 = 0 * a$.

Proof.

$$a*(1+(-1)) = (a*1)+(a*(-1))=0.$$

Qed.

Lemma MulMinOne. $(-1) * a = -a = a * -1$.

Proof.

$$a+(-1 * a) = (1*a)+(-1 * a) = 0.$$

Qed.

Lemma IntCanc.

$$c \neq 0 \wedge a * c = b * c \Rightarrow a = b.$$

Proof.

$$\text{Assume } c \neq 0 \wedge a * c = b * c.$$

$$(1) (a + (-b)) * c = (a * c) + ((-b) * c) = 0.$$

Therefore $a - b = 0$ (by ZeroDiv, 1).

Qed.

Let a is nonzero stand for $a \neq 0$.

Let p, q stand for nonzero integers.

[synonym divisor/-s] [synonym divide/-s]

Definition Divisor. A divisor of b is a nonzero integer a
such that for some n ($a * n = b$).

Let a divides b stand for a is a divisor of b .

Let $a \mid b$ stand for a is a divisor of b .

Lemma DivPlus.

$$q \mid a \wedge q \mid b \Rightarrow q \mid (a + b).$$

Definition EquMod. $a = b \pmod{q}$ iff $q \mid a - b$.

Definition NeqMod. $a \neq b \pmod{q}$ iff not ($a = b \pmod{q}$).

Lemma EquModRef. $a = a \pmod{q}$.

[ontored on]

Lemma EquModSym. $a = b \pmod{q} \Rightarrow b = a \pmod{q}$.

Proof.

Assume that $a = b \pmod{q}$.

$$(1) \text{ Take } n \text{ such that } q * n = a - b.$$

```

q * -n . = (-1) * (q * n) (by MulMinOne, MulAsso, MulComm, MulBubble)
      . = (-1) * (a - b) (by 1).
Therefore q | b-a.
qed.

```

Lemma EquModTrn. $a = b \pmod{q} \wedge b = c \pmod{q} \Rightarrow a = c \pmod{q}$.

Proof.

```

Assume that a = b (mod q) /\ b = c (mod q).
Take n such that q * n = a - b.
Take m such that q * m = b - c.
We have q * (n + m) = a - c.
qed.

```

Lemma EquModMul. $a = b \pmod{p * q} \Rightarrow a = b \pmod{p} \wedge a = b \pmod{q}$.

Proof.

```

Assume that a = b (mod p * q).
Take m such that (p * q) * m = a - b.
We have p * (q * m) = a - b = q * (p * m).
qed.
[/ontored]

```

3.7 Primzahlen

Primzahlen.

Wichtigste Eigenschaft, das erste Axiom.

Signature Prime. a is prime is an atom.

Let a prime stand for a prime nonzero integer.

Axiom.

Let n be a natural number.

Let p be a prime.

Let k be a natural number.

If $k \mid p^n$ **then** $k = 1$ or $p \mid k$.

Axiom.

Let k be a natural number.

$k \neq 0 \Rightarrow p \mid p^k$.

3.8 Ein Lemma über Potenzen

Das folgende Lemma wird eine zentrale Rolle im Beweis des zweiten Teiles des Satzes einnehmen. Wir liefern hier einen Beweis, der sich durch ein Symmetrieargument vereinfachen

ließe.

Lemma DLogN.

Let p be a prime.

Let a, b be natural numbers.

If $n = (p^a)^*c \wedge n = (p^b)^*d$ and p does not divide c and p does not divide d **then** $a = b$.

Proof.

Assume $n = (p^a)^*c$ and $n = (p^b)^*d$ and p does not divide c and p does not divide d .

b is not smaller than a .

Proof by Contradiction.

Assume $b < a$.

We **have** $p^a = (p^{a-b})^*(p^b)$.

- (1) $(p^a)^*c = (p^b)^*d$.
- (2) $((p^{a-b})^*(p^b))^*c = (p^b)^*d$.
- (3) $((p^b)^*(p^{a-b}))^*c = (p^b)^*d$ (by 1, MulComm).
- (4) $(p^b)^*((p^{a-b})^*c) = (p^b)^*d$ (by 3, MulAsso).
- (5) $((p^{a-b})^*c)^*(p^b) = d^*(p^b)$ (by 4, MulComm).
- (6) $((p^{a-b})^*c) = d$ (by 5, IntCanc, PotNotZero).

$a-b \neq 0$.

p is a divisor of p^{a-b} .

p is a divisor of $((p^{a-b})^*c)$.

p does divide d .

p does not divide d .

Contradiction.

end.

a is not smaller than b .

Proof by Contradiction.

Assume $a < b$.

We **have** $p^b = (p^{b-a})^*(p^a)$.

- (1) $(p^b)^*d = (p^a)^*c$.
- (2) $((p^{b-a})^*(p^a))^*d = (p^a)^*c$.
- (3) $((p^a)^*(p^{b-a}))^*d = (p^a)^*c$ (by 1, MulComm).
- (4) $(p^a)^*((p^{b-a})^*d) = (p^a)^*c$ (by 3, MulAsso).
- (5) $((p^{b-a})^*d)^*(p^a) = c^*(p^a)$ (by 4, MulComm).
- (6) $((p^{b-a})^*d) = c$ (by 5, IntCanc, PotNotZero).

$b-a \neq 0$.

p is a divisor of $p^{(b-a)}$.
 p is a divisor of $((p^{(b-a)})^d)$.

p does divide c.
 p does not divide c.

Contradiction.

end.

Therefore the thesis.

qed.

3.9 Endliche Mengen

Endliche Mengen führen wir als abstrakten Begriff ein und fordern dann axiomatisch die bekannten Eigenschaften. Interessant sind hier die Axiome `cardUnion0-2` bezüglich der Vereinigung von Gruppen. Da wir die ganze Zahlen recht abstrakt eingeführt haben und insbesondere kein Induktionsaxiom haben, müssen wir diese drei Axiome hier separat fordern, obwohl sie auf Metaebene aus `cardUnion0` gefolgert werden können.

Signature.

A finite set is a set.

Axiom.

Let M be a finite set.

Let N be a subset of M.

N is a finite set.

Axiom.

Let f be a function such that $\text{Dom}(f)$ is a finite set.

$\text{Range}(f)$ is a finite set.

Axiom.

Let M, N be finite set.

$\text{Prod}(M, N)$ is a finite set.

Signature.

Let M be a finite set.

$\text{card}(M)$ is a natural number.

Axiom.

Let M be a finite set.

Let N be a subset of M.

If $\text{card}(M) = \text{card}(N)$ then $M = N$.

Axiom.

Let M be a set such that for all elements N of M N is a finite set.
 \neg M is a finite set.

Axiom.
Let N_1, N_2 be finite sets.
 $N_1 \setminus N_2$ is a finite set.

Axiom cardUnion0.
Let N_1, N_2 be finite sets.
If N_1 and N_2 are disjoint then $\text{card}(N_1 \setminus N_2) = \text{card}(N_1) + \text{card}(N_2)$.

Axiom cardUnion1.
Let M be a set.
Let N be an element of M .
If M is a finite set such that every element of M is a finite set
and M is a disjoint collection
and for all elements N_1, N_2 of M $\text{card}(N_1) = \text{card}(N_2)$
then $\text{card}(\bigcup M) = \text{card}(N) * \text{card}(M)$.

Axiom cardUnion2.
Let M be a set.
Let k be an integer.
If M is a finite set such that every element of M is a finite set
and M is disjoint collection
and for all elements N of M $k \mid \text{card}(N)$
then $k \mid \text{card}(\bigcup M)$.

Axiom.
Let N_1, N_2 be finite sets.
 $\text{card}(N_1) = \text{card}(N_2)$ iff there is a function f such that (f is from N_1 to N_2 and f is
injective and f is surjective onto N_2).

Axiom.
Let M be a finite set.
If $\text{card}(M) \neq 0$ then M is not empty.

Axiom.
Let M be a finite set.
 $\text{card}(M) = 1$ iff there is $y \in M$ such that for all $x \in M$ $x = y$.

Axiom.
Let M be a finite set.
Assume $1 < \text{card}(M)$.
Let x be an element of M .
There is $y \in M$ such that $x \neq y$.

3.10 Der Satz von Lagrange

Der Satz von Lagrange ist ein grundlegender Satz der endlichen Gruppentheorie.

Definition.

A finite group is a group G such that $\text{El}(G)$ is a finite set.

Lemma.

Let G be a finite group.

Let U be a subgroup of G .

$\text{Cosets}(U, G)$ is a finite set.

Proof.

Define $f[g] = \text{Coset}(g, U, G)$ for g in $\text{El}(G)$.

$\text{Cosets}(U, G)$ is a subset of $\text{Range}(f)$.

Therefore the thesis.

Qed.

Definition.

Let G be a finite group.

Let U be a subgroup of G .

$\text{Index}(G, U) = \text{card}(\text{Cosets}(U, G))$.

Lemma.

Let G be a finite group.

Let U, V be subgroups of G such that V and U are conjugates in G .

$\text{card}(U) = \text{card}(V)$.

Proof.

Take an element g of $\text{El}(G)$ such that $V = \text{Conjugate}(g, U, G)$.

Define $f[u] = g \cdot \{u\} \cdot \text{Inv}(g, G)$ for u in U .

Let us show that f is from U to V .

$\text{Dom}(f) = U$.

Let us show that for all elements u of U $f[u]$ is an element of V .

Let u be an element of U .

$f[u]$ is an element of V .

end.

end.

Let us show that f is injective.

Let us show that for all elements u_1, u_2 of U if $f[u_1] = f[u_2]$ then $u_1 = u_2$.

Let u_1, u_2 be elements of U such that $f[u_1] = f[u_2]$.

We have $u_1 = (\text{Inv}(g, G) \cdot \{g\} \cdot (g \cdot \{u_1\} \cdot \text{Inv}(g, G))) \cdot \{g\}$.

We have $u_2 = (\text{Inv}(g, G) \cdot \{g\} \cdot (g \cdot \{u_2\} \cdot \text{Inv}(g, G))) \cdot \{g\}$.

Therefore $u_1 = (\text{Inv}(g, G) \cdot \{g\} \cdot f[u_1]) \cdot \{g\} = (\text{Inv}(g, G) \cdot \{g\} \cdot f[u_2]) \cdot \{g\} = u_2$.

end.

end.

Let us show that f is surjective onto V .

Let us show that for every element v of V there is an element u of U such that $f[u] = v$.

Let v be an element of V .

We can take an element u of U such that $v = (g \cdot u) \cdot \text{Inv}(g, G)$.

Hence $v = f[u]$.

end.

end.

qed.

Theorem Lagrange.

Let G be a finite group.

Let U be a subgroup of G .

$\text{card}(\text{El}(G)) = \text{card}(U) \cdot \text{card}(\text{Cosets}(U, G))$.

Proof.

Let us show that for all elements g of $\text{El}(G)$ $\text{card}(\text{Coset}(g, U, G)) = \text{card}(U)$.

Let g be an element of $\text{El}(G)$.

Define $f[u] = g \cdot u$ for u in U .

f is from U to $\text{Coset}(g, U, G)$.

f is injective.

Proof.

Let us show that for all elements u_1, u_2 of U If $f[u_1] = f[u_2]$ then $u_1 = u_2$.

Let u_1, u_2 be elements of U such that $f[u_1] = f[u_2]$.

We have $u_1 = \text{Inv}(g, G) \cdot (g \cdot u_1) = \text{Inv}(g, G) \cdot (g \cdot u_2) = u_2$.

Thus $u_1 = u_2$.

end.

Therefore the thesis.

end.

f is surjective onto $\text{Coset}(g, U, G)$.

Proof.

Let us show that for every element y of $\text{Coset}(g, U, G)$ there is an element u of U such that $f[u] = y$.

Let y be an element of $\text{Coset}(g, U, G)$.

Take an element u of U such that $y = g \cdot u$.

Then $f[u] = y$.

end.

Therefore the thesis.

end.

end.

- (1) $\text{Cosets}(U, G)$ is a disjoint collection and for all elements N_1, N_2 of $\text{Cosets}(U, G)$ $\text{card}(N_1) = \text{card}(N_2)$.
- (2) $\text{Cosets}(U, G)$ is a finite set such that for all element N_1 of $\text{Cosets}(U, G)$ N_1 is a finite set.
- (3) U is an element of $\text{Cosets}(U, G)$.

Therefore $\text{card}(\bigcup \text{Cosets}(U, G)) = \text{card}(U) \cdot \text{card}(\text{Cosets}(U, G))$ (by cardUnion , 1, 2, 3)

Qed.

3.11 Gruppenaktionen

Gruppenaktionen betrachten wir als Funktionen von $\text{Prod}(\text{El}(G), M)$ nach M mit zusätzlichen Eigenschaften.

Definition.

Let M be a set.

Let G be a group.

A group action from G on M is a function f

such that f is from $\text{Prod}(\text{El}(G), M)$ to M

and (for every element x of M $f[(\text{One}(G), x)] = x$)

and for every element x of M for all elements a, b of $\text{El}(G)$

$f[(a \cdot b, x)] = f[a, f[b, x]]$.

Definition.

Let M be a set.

Let G be a group.

Let f be a function from $\text{Prod}(\text{El}(G), M)$ to M .

Let x be an element of M .

$\text{Orbit}(x, f, M, G) = \{ f[a, x] \mid a \in \text{El}(G) \}$.

Definition.

Let M be a set.

Let G be a group.

Let f be a group action from G on M .

Let $x \in M$.

$\text{Stab}(x, f, M, G) = \{ y \mid y \in \text{El}(G) \text{ and } f[y, x] = x \}$.

Lemma.

Let M be a set.

Let G be a group.

Let f be a group action from G on M .

Let $x \in M$.

$\text{Stab}(x, f, M, G)$ is a subgroup of G .

Proof.

$\text{One}(G)$ is an element of $\text{Stab}(x, f, M, G)$.

Let us show that for all elements y, z of $\text{Stab}(x, f, M, G)$ $z \cdot y \in \text{Stab}(x, f, M, G)$.

Let y, z be elements of $\text{Stab}(x, f, M, G)$.

We have $f[(z \cdot^{G} \text{Inv}(y, G), x)] = x$.
Therefore $z \cdot^{G} \text{Inv}(y, G)$ is an element of $\text{Stab}(x, f, M, G)$.
end.

Therefore the thesis.
Qed.

Lemma.

Let M be a set.
Let G be a group.
Let f be a group action from G on M .
Let $x \ll M$.
Let g, h be elements of $\text{El}(G)$.
If $\text{Coset}(g, \text{Stab}(x, f, M, G), G) = \text{Coset}(h, \text{Stab}(x, f, M, G), G)$ then $f[(g, x)] = f[(h, x)]$.

3.12 Stabilisator und Orbit

Außerhalb von Beweisen lassen sich in Naproche keine Funktionen definieren. In der herkömmlichen Mathematik werden Funktionen oft definiert und erst nachträglich auf "Wohldefiniertheit" bzw. Repräsentantenunabhängigkeit überprüft. Man könnte diesen Vorgang über einen Relationsbegriff und weitere Axiome formalisieren. In diesem Text belassen wir es jedoch dabei, auf Metaebene begründet durch das vorangegangene Lemma, die Existenz der folgenden Funktion als Axiom zu fordern.

Axiom.

Let M be a set.
Let G be a group.
Let f be a group action from G on M .
Let $x \ll M$.
There is a function h such that
 h is from $\text{Cosets}(\text{Stab}(x, f, M, G), G)$ to $\text{Orbit}(x, f, M, G)$
and (for all elements i of $\text{El}(G)$ $h[\text{Coset}(i, \text{Stab}(x, f, M, G), G)] = f[(i, x)]$).

Lemma.

Let G be a finite group.
Let f be a group action from G on M .
Let $x \ll M$.
 $\text{Orbit}(x, f, M, G)$ is a finite set.

Proof.

Define $h[g] = f[(g, x)]$ for g in $\text{El}(G)$.
 $\text{Dom}(h)$ is a finite set.
 $\text{Orbit}(x, f, M, G)$ is a subset of $\text{Range}(h)$.

Proof.

Let us show that every element of $\text{Orbit}(x, f, M, G)$ is an element of $\text{Range}(h)$.

(1) Let y be an element of $\text{Orbit}(x, f, M, G)$.

We can take an element g_1 of $\text{El}(G)$ such that $y = f[(g_1, x)]$ (by 1).

Thus y is an element of $\text{Range}(h)$.

end.

end.

Therefore $\text{Orbit}(x, f, M, G)$ is a finite set.

Qed.

Lemma.

Let M be a set.

Let G be a finite group.

Let f be a group action from G on M .

Let $x \in M$.

$\text{Index}(G, \text{Stab}(x, f, M, G)) = \text{card}(\text{Orbit}(x, f, M, G))$.

Proof.

Take a function h such that

h is from $\text{Cosets}(\text{Stab}(x, f, M, G), G)$ to $\text{Orbit}(x, f, M, G)$

and (for all elements i of $\text{El}(G)$ $h[\text{Coset}(i, \text{Stab}(x, f, M, G), G)] = f[(i, x)]$).

h is surjective onto $\text{Orbit}(x, f, M, G)$.

Proof.

Let us show that for every element y of $\text{Orbit}(x, f, M, G)$ there is an element z of $\text{Dom}(f)$ such that $f[z] = y$.

Let y be an element of $\text{Orbit}(x, f, M, G)$.

Take an element i of $\text{El}(G)$ such that $f[(i, x)] = y$.

Then we have $h[\text{Coset}(i, \text{Stab}(x, f, M, G), G)] = y$.

end.

end.

h is injective.

Proof.

Let us show that for all elements h_1, h_2 of $\text{Cosets}(\text{Stab}(x, f, M, G), G)$ if $h[h_1] = h[h_2]$ then $h_1 = h_2$.

Let h_1, h_2 be elements of $\text{Cosets}(\text{Stab}(x, f, M, G), G)$ such that $h[h_1] = h[h_2]$.

Take elements g_1, g_2 of $\text{El}(G)$ such that $h_1 = \text{Coset}(g_1, \text{Stab}(x, f, M, G), G)$ and $h_2 = \text{Coset}(g_2, \text{Stab}(x, f, M, G), G)$.

Then $f[(g_1, x)] = f[(g_2, x)]$.

$f[(\text{Inv}(g_2, G) * \{G\} g_1), x] = f[(\text{Inv}(g_2, G) * \{G\} g_2), x] = x$.

Thus $\text{Inv}(g_2, G) * \{G\} g_1$ is an element of $\text{Stab}(x, f, M, G)$.

Therefore $h_1 = h_2$.

end.

end.

qed.

3.13 Fixpunkte

Wir führen den Begriff des Fixpunktes und der Menge der Fixpunkte einer Gruppenaktion ein und beweisen ein Lemma.

Definition.

Let M be a set.

Let G be a group.

Let f be a group action from G on M .

A fixed point of f on M on G is an element y of M such that for every element a of $\text{El}(G)$ $f[(a, y)] = y$.

Definition.

Let M be a set.

Let G be a group.

Let f be a group action from G on M .

$\text{fixedPoints}(f, M, G) = \{y \mid y \text{ is a fixed point of } f \text{ on } M \text{ on } G\}$.

Lemma.

Let M be a finite set.

Let G be a group.

Let f be a group action from G on M .

$\text{fixedPoints}(f, M, G)$ is a finite set.

Proof.

Let us show that every fixed point of f on M on G is an element of M .

Let x be a fixed point of f on M on G .

Then x is an element of M .

end.

$\text{fixedPoints}(f, M, G)$ is a subset of M .

Therefore the thesis.

Qed.

Lemma.

Let M be a set.

Let G be a finite group.

Let f be a group action from G on M .

Let x be an element of M .

x is a fixed point of f on M on G iff $\text{card}(\text{Orbit}(x, f, M, G)) = 1$.

Lemma OrbitsIntersect.

Let M be a set.

Let G be a group.

Let f be a group action from G on M .

Let x_1, x_2 be elements of M such that $\text{Orbit}(x_1, f, M, G)$ and $\text{Orbit}(x_2, f, M, G)$ are not disjoint.

x_1 is an element of $\text{Orbit}(x_2, f, M, G)$.

Proof.

Take $y \ll M$ such that $y \ll \text{Orbit}(x_1, f, M, G)$ and $y \ll \text{Orbit}(x_2, f, M, G)$.

Take $g_1 \ll \text{El}(G)$ such that $f[(g_1, x_1)] = y$.

Take $g_2 \ll \text{El}(G)$ such that $f[(g_2, x_2)] = y$.

$$x_1 = f[(\text{Inv}(g_1, G) \cdot \{G\} g_1, x_1)] = f[(\text{Inv}(g_1, G), y)] = f[(\text{Inv}(g_1, G) \cdot \{G\} g_2, x_2)].$$

Therefore the thesis.

Qed.

Lemma.

Let M be a set.

Let G be a group.

Let f be a group action from G on M .

Let x_1, x_2 be elements of M such that $\text{Orbit}(x_1, f, M, G)$ and $\text{Orbit}(x_2, f, M, G)$ are not disjoint.

$\text{Orbit}(x_1, f, M, G) = \text{Orbit}(x_2, f, M, G)$.

Proof.

Let us show that every element of $\text{Orbit}(x_1, f, M, G)$ is an element of $\text{Orbit}(x_2, f, M, G)$.

Let x_3 be an element of $\text{Orbit}(x_1, f, M, G)$.

x_1 is an element of $\text{Orbit}(x_2, f, M, G)$ (by `OrbitsIntersect`).

Thus x_3 is an element of $\text{Orbit}(x_2, f, M, G)$.

end.

Let us show that every element of $\text{Orbit}(x_2, f, M, G)$ is an element of $\text{Orbit}(x_1, f, M, G)$.

Let x_3 be an element of $\text{Orbit}(x_2, f, M, G)$.

x_2 is an element of $\text{Orbit}(x_1, f, M, G)$ (by `OrbitsIntersect`).

Thus x_3 is an element of $\text{Orbit}(x_1, f, M, G)$.

end.

Qed.

Definition.

Let M be a set.

Let G be a group.

Let f be a group action from G on M .

$\text{OrbitsNotFix}(f, M, G) = \{\text{Orbit}(x, f, M, G) \mid x \text{ is an element of } M \text{ and } x \text{ is not a fixed point of } f \text{ on } M \text{ on } G\}$.

Lemma.

Let M be a set.

Let G be a group.

Let f be a group action from G on M .

$\text{OrbitsNotFix}(f, M, G)$ is a disjoint collection.

Lemma.

Let M be a set.

Let G be a group.
 Let f be a group action from G on M .
 $\text{OrbitsNotFix}(f, M, G) = M \setminus \text{fixedPoints}(f, M, G)$.
Proof.
 Let us show that every element of $\text{OrbitsNotFix}(f, M, G)$ is an element of $M \setminus \text{fixedPoints}(f, M, G)$.
 Let x be an element of $\text{OrbitsNotFix}(f, M, G)$.
 Take an element y of M such that x is an element of $\text{Orbit}(y, f, M, G)$ and y is not an element of $\text{fixedPoints}(f, M, G)$.
 x is an element of M .
 x is not a fixed point of f on M on G .
 $\text{fixedPoints}(f, M, G)$ is a subset of M .
 Therefore x is an element of $M \setminus \text{fixedPoints}(f, M, G)$.
end.

Let us show that every element of $M \setminus \text{fixedPoints}(f, M, G)$ is an element of $\text{OrbitsNotFix}(f, M, G)$.
 Let x be an element of $M \setminus \text{fixedPoints}(f, M, G)$.
 x is an element of M .
 x is not a fixed point of f on M on G .
 $\text{Orbit}(x, f, M, G)$ is an element of $\text{OrbitsNotFix}(f, M, G)$.
 x is an element of $\text{Orbit}(x, f, M, G)$.
 Therefore x is an element of $\text{OrbitsNotFix}(f, M, G)$.
end.
qed.

Lemma.
 Let M be a set.
 Let G be a group.
 Let f be a group action from G on M .
 $\text{OrbitsNotFix}(f, M, G)$ and $\text{fixedPoints}(f, M, G)$ are disjoint.

Lemma.
 Let M be a set.
 Let G be a group.
 Let f be a group action from G on M .
 $(\text{OrbitsNotFix}(f, M, G) \cup \text{fixedPoints}(f, M, G)) = M$.

Lemma.
 Let M be a finite set.
 Let G be a finite group.
 Let f be a group action from G on M .
 $\text{OrbitsNotFix}(f, M, G)$ is a finite set.

Proof.
 Define $h[x] = \text{Orbit}(x, f, M, G)$ for x in M .
 $\text{OrbitsNotFix}(f, M, G)$ is a subset of $\text{Range}(h)$.
Qed.

Lemma.

Let M be a finite set.

Let G be a finite group.

Let f be a group action from G on M .

$\text{card}(M) = \text{card}(\text{OrbitsNotFix}(f, M, G)) + \text{card}(\text{fixedPoints}(f, M, G))$
 $= \text{card}(\text{fixedPoints}(f, M, G)) + \text{card}(\text{OrbitsNotFix}(f, M, G)).$

Signature.

Let p be a prime.

A group of order p is a finite group H such that
(there is a natural number n such that $\text{card}(\text{El}(H)) = p^n$).

Lemma.

Let M be a finite set.

Let p be a prime.

Let G be a group of order p .

Let f be a group action from G on M .

$\text{card}(\text{fixedPoints}(f, M, G)) = \text{card}(M) \pmod{p}$.

Proof.

$\text{OrbitsNotFix}(f, M, G)$ is a subset of M .

Let us show that $p \mid \text{card}(\text{OrbitsNotFix}(f, M, G))$.

Let us show that for all elements N_1 of $\text{OrbitsNotFix}(f, M, G)$ $p \mid \text{card}(N_1)$.

Let N be an element of $\text{OrbitsNotFix}(f, M, G)$.

Take an element x of M such that $N = \text{Orbit}(x, f, M, G)$.

Let us show that $\text{card}(N) \neq 1$.

Assume the contrary.

x is not a fixed point of f on M on G .

x is an element of N .

Thus x is a fixed point of f on M on G .

Contradiction.

end.

We have $\text{card}(N) = \text{Index}(G, \text{Stab}(x, f, M, G))$.

Hence $\text{card}(\text{El}(G)) = \text{card}(\text{Stab}(x, f, M, G)) \cdot \text{card}(N)$ and $\text{card}(N) \mid \text{card}(\text{El}(G))$.

Therefore $p \mid \text{card}(N)$.

end.

(1) $\text{OrbitsNotFix}(f, M, G)$ is a finite set such that every element of $\text{OrbitsNotFix}(f, M, G)$ is a finite set.

(2) $\text{OrbitsNotFix}(f, M, G)$ is a disjoint collection and for all elements N of $\text{OrbitsNotFix}(f, M, G)$ $p \mid \text{card}(N)$.

Therefore $p \mid \text{card}(\setminus \text{OrbitsNotFix}(f, M, G))$ (by cardUnion2 , 1, 2).
end.

We have $\text{card}(M) = \text{card}(\text{fixedPoints}(f, M, G)) + \text{card}(\setminus \text{OrbitsNotFix}(f, M, G))$.

Therefore $\text{card}(M) = \text{card}(\text{fixedPoints}(f, M, G)) \pmod{p}$.
qed.

3.14 Eine Gruppenaktion

Um den zweiten Satz von Sylow zu beweisen, muss eine explizite Gruppenaktion definiert werden.

Im Gegensatz zur abstrakten Formulierung des Gruppen- und Zahlbegriffes in dieser Formalisierung lässt sich aufgrund der Formulierung der Definition von Gruppenaktion nun in Naproche zeigen, dass es sich bei dieser Funktion um eine Gruppenaktion handelt.

Lemma.

Let P be a subgroup of G .

Let U be a subgroup of G .

Let u be an element of U .

Let x, y be elements of $\text{El}(G)$ such that $\text{Coset}(x, P, G) = \text{Coset}(y, P, G)$.

Every element of $\text{Coset}(u^{*G} x, P, G)$ is an element of $\text{Coset}(u^{*G} y, P, G)$.

Proof.

Let i be an element of $\text{Coset}(u^{*G} x, P, G)$.

Take an element p of P such that $i = (u^{*G} x)^{*G} p$.

Then we have $\text{Inv}(u, G)^{*G} i = \text{Inv}(u, G)^{*G} ((u^{*G} x)^{*G} p)$

$= ((\text{Inv}(u, G)^{*G} u)^{*G} x)^{*G} p$

$= x^{*G} p$.

$\text{Inv}(u, G)^{*G} i$ is an element of $\text{Coset}(x, P, G)$.

Therefore $\text{Inv}(u, G)^{*G} i$ is an element of $\text{Coset}(y, P, G)$

and i is an element of $\text{Coset}(u^{*G} y, P, G)$.

qed.

Lemma.

Let P be a subgroup of G .

Let U be a subgroup of G .

Let u be an element of U .

Let x, y be elements of $\text{El}(G)$ such that $\text{Coset}(x, P, G) = \text{Coset}(y, P, G)$.

$\text{Coset}(u^{*G} x, P, G) = \text{Coset}(u^{*G} y, P, G)$.

Proof.

Every element of $\text{Coset}(u^{*G} x, P, G)$ is an element of $\text{Coset}(u^{*G} y, P, G)$.

Every element of $\text{Coset}(u^{*G} y, P, G)$ is an element of $\text{Coset}(u^{*G} x, P, G)$.

Therefore the thesis.

Qed.

Aufgrund des vorangegangenen Lemmas führt die folgende Definition nicht zu Problemen hinsichtlich der Konsistenz der Formalisierung (Siehe 3.12).

Definition.

Let P be a subgroup of G .

Let U be a subgroup of G .

$\text{Op}(U, P, G)$ is a function f

such that f is from $\text{Prod}(\text{El}(\text{Gr}(U, G)), \text{Cosets}(P, G))$ to $\text{Cosets}(P, G)$ and

for all elements u of U for all elements x of $\text{El}(G)$

$f[(u, \text{Coset}(x, P, G))] = \text{Coset}(u *^{\{G\}} x, P, G)$.

Lemma.

Let P be a subgroup of G .

Let U be a subgroup of G .

$\text{Op}(U, P, G)$ is a group action from $\text{Gr}(U, G)$ on $\text{Cosets}(P, G)$.

Proof.

Take a function f such that $f = \text{Op}(U, P, G)$.

Take a group H such that $H = \text{Gr}(U, G)$.

Take a set M such that $M = \text{Cosets}(P, G)$.

For every element x of M we have $f[(\text{One}(H), x)] = x$.

Let us show that for every element x of M for all elements a, b of $\text{El}(H)$

$f[(a *^{\{H\}} b, x)] = f[(a, f[(b, x)])]$.

Let x be an element of M .

Let a, b be elements of $\text{El}(H)$.

Take an element g of $\text{El}(G)$ such that $x = \text{Coset}(g, P, G)$.

We have $f[(b, x)] = \text{Coset}(b *^{\{G\}} g, P, G)$.

$f[(a, f[(b, x)])] = \text{Coset}(a *^{\{G\}} (b *^{\{G\}} g), P, G)$.

$f[(a *^{\{H\}} b, x)] = \text{Coset}((a *^{\{G\}} b) *^{\{G\}} g, P, G)$.

Thus $f[(a, f[(b, x)])] = f[(a *^{\{H\}} b, x)]$.

end.

Therefore the thesis.

Qed.

3.15 Vorbereitung für den zweiten Satz von Sylow

Wir führen die Begriffe des zweiten Satzes von Sylow ein und zeigen, dass zwei Sylowgruppen die gleiche Kardinalität haben.

Signature.

Let p be a prime.

A subgroup of G of order p is a subgroup U of G such that $\text{Gr}(U, G)$ is a group of order p .

Definition.

Let p be a prime.

Let G be a finite group.

$\text{Syl}(p, G) = \{P \mid P \text{ is a subgroup of } G \text{ of order } p \text{ and } \text{not } (p \mid \text{Index}(G, P))\}$.

Lemma SylSize.

Let p be a prime.

Let G be a finite group.

Let P, U be elements of $\text{Syl}(p, G)$.

$\text{card}(U) = \text{card}(P)$.

Proof.

Take natural numbers n, m such that $p^n = \text{card}(U)$ and $p^m = \text{card}(P)$.

(1) $\text{card}(\text{El}(G)) = (p^n) * \text{Index}(G, U)$
 and $\text{card}(\text{El}(G)) = (p^m) * \text{Index}(G, P)$
 and p does not divide $\text{Index}(G, U)$
 and p does not divide $\text{Index}(G, P)$.

Thus we have $n = m$ (by $\text{DLogN}, 1$).

Therefore the thesis.

Qed.

3.16 Der zweite Satz von Sylow

Schlussendlich können wir den Beweis des zweiten Satzes von Sylow in Naproche formulieren.

Theorem Sylow2a.

Let p be a prime.

Let G be a finite group.

Let P be an element of $\text{Syl}(p, G)$.

Let U be a subgroup of G of order p .

There is an element g of $\text{El}(G)$ such that $\text{Conjugate}(g, U, G)$ is a subset of P .

Proof.

Take a group action f from $\text{Gr}(U, G)$ on $\text{Cosets}(P, G)$ such that $f = \text{Op}(U, P, G)$.

Let us show that $\text{card}(\text{fixedPoints}(f, \text{Cosets}(P, G), \text{Gr}(U, G))) \neq 0$.

We have $\text{card}(\text{fixedPoints}(f, \text{Cosets}(P, G), \text{Gr}(U, G))) = \text{Index}(G, P) \pmod{p}$.

p does not divide $\text{Index}(G, P)$.

Therefore $\text{Index}(G, P) \neq 0 \pmod{p}$.

end.

We can take an element x of $\text{fixedPoints}(f, \text{Cosets}(P, G), \text{Gr}(U, G))$
and an element g of $\text{El}(G)$ such that $x = \text{Coset}(g, P, G)$.

Let us show that every element of $\text{Conjugate}(\text{Inv}(g, G), U, G)$ is an element of P .
Let h be an element of $\text{Conjugate}(\text{Inv}(g, G), U, G)$.

Take an element u of U such that $h = \text{Inv}(g, G) \cdot (u \cdot g)$.

We have $\text{Coset}(g, P, G) = f[(u, x)] = \text{Coset}(u \cdot g, P, G)$.

Therefore $\text{Inv}(g, G) \cdot (u \cdot g)$ is an element of P .

Thus h is an element of P .

end.

Qed.

Theorem Sylow2b.

Let p be a prime.

Let G be a finite group.

Let P, U be elements of $\text{Syl}(p, G)$.

P and U are conjugates in G .

Proof.

Take an element g of $\text{El}(G)$ such that $\text{Conjugate}(g, U, G)$ is a subset of P .

$\text{card}(\text{Conjugate}(g, U, G)) = \text{card}(U) = \text{card}(P)$.

Hence $\text{Conjugate}(g, U, G) = P$.

qed.

4 Vergleich

4.1 Direkte Gegenüberstellung

4.1.1 Herkömmliche Mathematik

Theorem 4.1. *Sei p eine Primzahl und G eine endliche Gruppe mit $|G| = p^r \cdot m$, sodass $p \nmid m$. Sei $U \leq G$ eine p -Untergruppe, und sei $P \leq G$ eine p -Sylowgruppe. Dann gilt:*

(i) *Es gibt ein $g \in G$ mit*

$$gUg^{-1} \subseteq P.$$

(ii) *Je zwei p -Sylowgruppen sind konjugiert.*

Beweis. (i) Setze $X := G/P$ und betrachte die Gruppenaktion

$$\phi : U \times X \rightarrow X$$

$$(u, gP) \mapsto ugP.$$

Nach Lemma 1.5 gilt

$$|X| = [G : P] = m \equiv |X^G| \pmod{p}.$$

Da $p \nmid m$, gilt $X^G \neq \emptyset$. Es existiert also ein $g \in G$, sodass $ugP = gP$ für alle $u \in U$. Folglich ist $g^{-1}Ug \subseteq P$ und damit U konjugiert zu P bzgl. g^{-1} .

(ii) Dies gilt insbesondere im Falle $U \in \text{Syl}_p(G)$. □

4.1.2 Lean

```
lemma sylow_2 [fintype G] {p : ℕ} (hp : nat.prime p)
(H K : set G) [is_sylow H hp] [is_sylow K hp] :
  ∃ g : G, H = conjugate_set g K :=

have hs : card (left_cosets K) = card G / (p ^ dlogn p (card G)) :=
  (nat.mul_right_inj (pos_pow_of_pos (dlogn p (card G)) hp.pos)).1
$ by rw [← card_sylow K hp, ← card_eq_card_cosets_mul_card_subgroup,
  card_sylow K hp,
  nat.div_mul_cancel (dlogn_dvd _ hp.1)],

have hmodeq : card G / (p ^ dlogn p (card G)) ≡ card (fixed_points H (
  left_cosets K)) [MOD p] :=
eq.subst hs (mul_action.card_modeq_card_fixed_points hp (card_sylow H
  hp)),

have hfixed : 0 < card (fixed_points H (left_cosets K)) :=
```

```

nat.pos_of_ne_zero
( $\lambda$  h, (not_dvd_div_dlogn (fintype.card_pos_iff.2  $\langle$ (1 : G) $\rangle$ ) hp.1)
  (by rwa [h, nat.modeq.modeq_zero_iff] at hmodeq)),

```

4.1.3 Naproche

Theorem Sylow2a.

Let p be a prime.

Let G be a finite group.

Let P be an element of $\text{Syl}(p, G)$.

Let U be a subgroup of G of order p .

There is an element g of $\text{El}(G)$ such that $\text{Conjugate}(g, U, G)$ is a subset of P .

Proof.

Take a group action f from $\text{Gr}(U, G)$ on $\text{Cosets}(P, G)$ such that $f = \text{Op}(U, P, G)$.

Let us show that $\text{card}(\text{fixedPoints}(f, \text{Cosets}(P, G), \text{Gr}(U, G))) \neq 0$.

We have $\text{card}(\text{fixedPoints}(f, \text{Cosets}(P, G), \text{Gr}(U, G))) = \text{Index}(G, P) \pmod{p}$.

p does not divide $\text{Index}(G, P)$.

Therefore $\text{Index}(G, P) \neq 0 \pmod{p}$.

end.

We can take an element x of $\text{fixedPoints}(f, \text{Cosets}(P, G), \text{Gr}(U, G))$

and an element g of $\text{El}(G)$ such that $x = \text{Coset}(g, P, G)$.

Let us show that every element of $\text{Conjugate}(\text{Inv}(g, G), U, G)$ is an element of P .

Let h be an element of $\text{Conjugate}(\text{Inv}(g, G), U, G)$.

Take an element u of U such that $h = \text{Inv}(g, G) *^{\{G\}} (u *^{\{G\}} g)$.

We have $\text{Coset}(g, P, G) = f[(u, x)] = \text{Coset}((u *^{\{G\}} g), P, G)$.

Therefore $\text{Inv}(g, G) *^{\{G\}} (u *^{\{G\}} g)$ is an element of P .

Thus h is an element of P .

end.

Qed.

Theorem Sylow2b.

Let p be a prime.

Let G be a finite group.

Let P, U be elements of $\text{Syl}(p, G)$.

P and U are conjugates in G .

Proof.

Take an element g of $\text{El}(G)$ such that $\text{Conjugate}(g, U, G)$ is a subset of P .

$\text{card}(\text{Conjugate}(g, U, G)) = \text{card}(U) = \text{card}(P)$.

Hence $\text{Conjugate}(g, U, G) = P$.

qed.

4.2 Beweisführung

Der zentrale Unterschied zwischen den beiden Formalisierungen ist, dass Lean eine funktionale Programmiersprache ist und auf der Typentheorie basiert. Dies führt zu einer sehr technischen Aufbau des Lean-Beweises durch Kombination von Lambda-Termen.

Im Gegensatz dazu arbeitet das Naprochesystem in einem deklarativen Stil, d.h es handelt sich um eine Aneinanderreihung von Aussagen, die durch den internen Reasoner sowie den externen ATP verifiziert werden.

Dies ermöglicht es dem Naproche-Text auf der einen Seite, auf struktureller Ebene recht nah am Stil der herkömmlichen Mathematik zu bleiben.

Auf der anderen Seite ermöglicht die Beschaffenheit von Lean als Programmiersprache hier jedoch gewisse argumentative Methoden wie Symmetrie durch sogenannte "Taktiken" umzusetzen, die sich im deklarativen Stil von Naproche momentan nicht abbilden lassen (Siehe den Beweis in Abschnitt 3.8).

```
example (x y : ℕ) (h : x = y) : y = x :=
begin
  revert x y,
  intros,
  symmetry,
  assumption
end
```

(Beispiel eines Taktik-Beweises in Lean unter Verwendung von `symmetry` [4])

4.3 Algebraische Strukturen

Ein weiterer Unterschied ist die Behandlung von algebraischen Strukturen hinsichtlich der Begriffsbildung.

So existiert in Lean das System der Typenklassen, während in Naproche keine bis auf einen naiven Mengen und Funktionsbegriff keine Strukturelemente bereitgestellt werden.

Während in der Formalisierung in Lean beispielsweise die Gruppen strukturell als Erweiterung des Typenklasse `Monoid` entstehen, werden diese in der Naproche-Formalisierung als abstrakte Notion eingeführt. Exemplarisch stellt sich hier die Behandlung der Untergruppen heraus: `instance subtype.group s : set α [is_subgroup] s : group s := ...`

Hier wird in Lean gezeigt, dass Untergruppen eine `instance` der Typenklasse `group` ist. Dies sorgt dafür, dass sich Untergruppen in Lean nun ohne weitere Notation auch als Gruppen auffassen lassen. Da Naproche keine solche Strukturen bereitstellt, benötigen wir hierfür die axiomatische geforderte Gruppe $\text{Gr}(U, G)$ (siehe Abschnitt 3.2).

4.4 Funktionsdefinitionen

Lean enthält als Funktionale Programmiersprache Möglichkeiten, Funktionen und Typklassen zu definieren und mit diesen zu arbeiten.

Das Naproche System stellt auch einen Mengen, Funktionen bereit, dieser ist jedoch sehr . So ist es zum Beispiel nicht möglich außerhalb . Dies führt zu den angesprochenen Problemen (Siehe).

4.5 Struktur

In einem herkömmlichen Beweis [1] des zweiten Sylowsatzes findet man auch die Unterteilung in Kapitel welche sich in der Dateistruktur unserer Formalisierung wiederfindet. Allerdings werden im herkömmlichen Beweis grundlegende Eigenschaften von endlichen Kardinalitäten, endlichen Mengen und den ganzen Zahlen als bekannt vorausgesetzt. Dies spiegelt sich in der Formalisierung in Naproche insofern wieder, als diese Eigenschaften axiomatisch gefordert und nicht explizit bewiesen werden, wie dies in der Lean Formalisierung praktiziert wird.

Es wäre wünschenswert, alle Dateien als Ganzes von Naproche verifizieren zu lassen. Hier scheitert das System, weswegen wir in jeder Datei jeweils die Aussagen und Definitionen der vorangegangenen Dateien als Axiome ohne Beweise einbinden.

Dieser Ansatz findet sich insofern in der herkömmlichen Mathematik, als dort auch nur die Aussagen eines bereits bewiesenen Theorems zitiert werden, wohingegen die exakte Ausgestaltung des Beweises meist eine untergeordnete Rolle spielt: "Nach Lemma 1.5 gilt [...]"[1].

Im Gegensatz hierzu profitiert Lean als Programmiersprache von der Möglichkeit Bibliotheken einzubinden. Es ist insbesondere möglich, Dateien oder ganze Bibliotheken durch `leanpkg build in .olean` Binärdateien zu übersetzen, die vom Lean dann weitaus schneller verifiziert werden können. Dies ermöglicht es, mit umfangreichen Bibliotheken wie der `mathlib` [3] interaktiv zu arbeiten.

4.6 Lesbarkeit

In der vorangegangenen Gegenüberstellung ist erkennbar, dass der deklarative Stil der Herangehensweise der herkömmlichen Mathematik ähnelt. So ist bis auf Layout- und Gestaltungsebene eine große oberflächliche Ähnlichkeit zwischen den Beweisen zu erkennen.

Betrachte:

Nach Lemma 1.5 gilt

$$|X| = [G : P] \equiv |X^G| \pmod{p}.$$

We `have` `card(fixedPoints(f, Cosets(P, G), Gr(U, G))) = Index(G, P) (mod p)`.

Im Gegensatz hierzu lässt sich die Formalisierung in Lean nur mit einer gewissen Vorkenntnis in funktionaler Programmierung und dem Beweisen mithilfe von Taktiken verstehen.

4.7 Overloading

Ein Vorteil von Lean gegenüber Naproche / ForTheL hinsichtlich der Lesbarkeit ist die Arbeit des "Elaborators". Aufgrund der strikten Typisierung ist Lean in der Lage, beispielsweise bei

Ausdrücken der Form $a * b$ mit $a, b : G$ selbstständig zu ermitteln, dass es sich bei $*$ um die Verknüpfung von G handelt. Da dies in Naproche nicht möglich ist, entstehen sperrige Ausdrücke wie beispielsweise `fixedPoints(Op(U, P, G), Cosets(P, G), Gr(U, G))`, die die Lesbarkeit des Naprochetextes vermindern. In der herkömmlichen Mathematik findet sich hier oft ein Kompromiss. So werden an einigen Stellen die Abhängigkeiten gekennzeichnet, wie zum Beispiel im Ausdruck $\text{Syl}_p(G)$, während an anderen Stellen, wie in *Je zwei p -Sylowgruppen sind konjugiert*, aus dem Kontext ersichtlich wird, welche Gruppe gemeint ist.

4.8 User Experience

Wir wollen nun einige Aspekte des Arbeitens mit den beiden Systemen beschreiben.

Ein Punkt ist hier die Stabilität der Systeme. Solange man die gleiche Lean Version verwendet, kann man einen einmal geschriebenen Leantext in verschiedenen Umgebungen verifizieren. Für verschiedene Versionen von Lean gilt dies jedoch nicht. So ist zum Beispiel die ursprüngliche Lean Datei [2] mit der Lean Version 3.4.2 nicht mehr kompatibel und musste deswegen angepasst werden. Im Gegensatz hierzu ist das Verifizieren in Naproche sehr instabil. So führt das Verändern weniger Zeilen im Quelltext häufig dazu, dass Aussagen an anderen Stellen, die vorher verifiziert wurden, nicht mehr verifiziert werden. Insbesondere ist hier das Zusammenarbeiten mehrerer Personen mit verschiedenen Betriebssystemen kritisch, da nach unserer Erfahrung unter Linux/Mac verifizierte Dateien unter Windows oftmals nicht ohne weitere Veränderungen verifiziert werden konnten.

Hier scheint insbesondere der Reasoner in Naproche eine Rolle zu spielen, der in unserer Arbeit große Schwierigkeiten hatte, Funktionsterme auf ontologische Korrektheit zu überprüfen, weswegen wir deren Einsatz auf ein Minimum reduzieren mussten.

Schlussendlich würden beide Systeme von einer ausgeprägteren Dokumentation profitieren, da momentan viel Zeit darauf verloren geht, sich über einen Trial-and-Error-Prozess mit den Systemen vertraut zu machen. Lean profitiert hier von den ausgeprägten Fehlermeldungen, die oftmals klar erkennbar machen, welche Zeile in welcher Form geändert werden müssen, um zu eine valide Datei zu erhalten. In Isabelle-Naproche

5 Diskussion

5.1 Formalisierungskonventionen

Es ist aufgefallen, so ermöglichen gemeinsame Definition von mathematischen Begriffen

Ein Beispiel hierfür ist die Implementierung der Nebenklassen. Hier nutzen verschiedene Ansätze, die jedoch auf Metaebene äquivalent sind.

interessant hier Möglichkeiten zu schaffen, andere Formalisierungen zu referenzieren / einzubinden.

Interessant in diesem Kontext ist das Projekt Formal Abstracts [5]

5.2 Vorschläge zur Weiterentwicklung an Naproche

Als Resultat des Formalisierungsprozesses sind einige Ideen für die Weiterentwicklung des Naprochesystems entstanden.

- So . Es wäre wünschenswert einen kanonischen Weg zu haben, in Naproche mit algebraischen Strukturen umzugehen. Auch einen Quotientenbegriff.
- Definitionen von Funktionen und ggf anderen mathematischen Objekten internalisieren.
- Verbesserung des Reasoners insbesondere hinsichtlich von Funktionenstermen. So wurde in einer früheren Version des Formalisierung des Sylowsatzes
- Für die Einbindung von anderen Dateien existiert bereits der `read` Befehl. Dieser kann allerdings in der momentanen Ausführung nur auf Dateien im `examples` Ordner zugreifen. Hier wäre es wünschenswert, auch auf andere Verzeichnisse zugreifen zu können.
- Hinsichtlich dem Einlesen von Dateien ist eine weitere Idee, eine Möglichkeit zu schaffen, bereits verifizierte Texte einzulesen, ohne sie erneut verifizieren zu müssen. Dies führt momentan dazu, dass
- Bibliotheken vorkompilieren: erster Schritt: `read` Befehl
- Exportieren von Theoremen / Definitionen ... Speichern des Beweises zum späteren schnellen Zugriff wie durch `olean` Dateien.
- Unterscheiden von Axiomen und Annahmen"
- Error Messages
- Betriebssysteme unterschiedliche Ergebnisse <- Überarbeiten (`read`, Funktionalität des Reasoners
- Dokumentation

5.3 Lesbarkeit als Ziel

Unsere Formalisierung (mit Latex Typesetting) sehr ähnlich zu Schröers Algebraskript. Der deklarative Stil ermöglicht dies. Lean in der Hinsicht anpassen.

Interessant, den Ansatz von Lean auf natürlicherer Sprachebene.

Literatur

- [1] J. Schröer. Eine Einführung in die Algebra (Skript, WS 19/20, Bonn). 2020.
- [2] C. Hughes. Sylow. <https://github.com/ChrisHughes24/Sylow>.
- [3] mathlib. <https://github.com/leanprover-community/mathlib>.
- [4] J. Avigad. Theorem Proving in Lean. https://leanprover.github.io/theorem_proving_in_lean/tactics.html.
- [5] T. Hales. Formal Abstracts. <https://formalabstracts.github.io>, 2017.